/+ dub.sdl:
    name "C"
    dependency "dcomp" version=">=0.3.2"
    lflags "-stack_size" "100000000"
+/
 
import std.algorithm, std.conv, std.range, std.stdio;
import std.functional;

// import dcomp.scanner;
 
long co = 0;
int calcBase(int n, int d) {
    co++;
    if (n == 1) return d;
    int ans = 0;
    foreach (i; 0..128) {
        ans = max(ans, calc(n-1, d^i) + i);
    }
    return ans;
}
 
struct memoInts(alias pred) {
    import std.traits : ReturnType, ParameterTypeTuple, isIntegral;
    import std.typecons : tuple, Tuple;
    import std.meta;
    alias R = ReturnType!pred;
    alias Args = ParameterTypeTuple!pred;
    static assert (allSatisfy!(isIntegral, Args)); // should be int only?

    static immutable N = Args.length;
    static string toTuple(string s) { //int[N] -> (N[0], N[1], N[2], ...)
        return "(" ~ iota(N).map!(i => s~"["~i.to!string~"]").join(",") ~ ")";
    }
    static string toArray(string s) { //int[N] -> [N[0]][N[1]][N[2]]...
        return iota(N).map!(i => "["~s~"["~i.to!string~"]]").join("");
    }
    template NArray(T, int N) {
        static if (!N) alias NArray = T;
        else alias NArray = NArray!(T, N-1)[];
    }
    int[2][N] rng;
    NArray!(R, N) dp;
    NArray!(bool, N) used;
    void init(int[2][N] rng) {
        this.rng = rng;
        int[N] len = rng[].map!(a => a[1]-a[0]+1).array;
        dp = mixin("new NArray!(R, N)"~toTuple("len"));
        used = mixin("new NArray!(bool, N)"~toTuple("len"));
    }
    R opCall(Args args) {
        int[N] idx;
        foreach (i, v; args) {
            assert(rng[i][0] <= v && v <= rng[i][1]);
            idx[i] = v - rng[i][0];
        }
        if (mixin("used"~toArray("idx"))) return mixin("dp"~toArray("idx"));
        mixin("used"~toArray("idx")) = true;
        auto r = pred(args); 
        mixin("dp"~toArray("idx")) = r;
        return r;
    }
}
memoInts!calcBase calc;
 
int main(string[] argv) {
    calc.init([[0, 1000], [0, 127]]);
    auto sc = new Scanner();
    int t;
    sc.read(t);

    foreach (unused; 0..t) {
        int n, d;
        sc.read(n, d);
        writeln(calc(n, d));
    }
    return 0;
}
/* IMPORT /Users/yosupo/Program/dcomp/source/dcomp/scanner.d */
// module dcomp.scanner;

class Scanner {
    import std.stdio : File, stdin;
    import std.conv : to;
    import std.range : front, popFront, array, ElementType;
    import std.array : split;
    import std.traits : isSomeChar, isStaticArray, isArray; 
    import std.algorithm : map;
    File f;
    this(File f = stdin) {
        this.f = f;
    }
    string[] buf;
    private bool succ() {
        while (!buf.length) {
            if (f.eof) return false;
            buf = f.readln.split;
        }
        return true;
    }
    private bool readSingle(T)(ref T x) {
        if (!succ()) return false;
        static if (isArray!T) {
            alias E = ElementType!T;
            static if (isSomeChar!E) {
                //string or char[10] etc
                x = buf.front;
                buf.popFront;
            } else {
                static if (isStaticArray!T) {
                    //static
                    assert(buf.length == T.length);
                }
                x = buf.map!(to!E).array;
                buf.length = 0;                
            }
        } else {
            x = buf.front.to!T;
            buf.popFront;            
        }
        return true;
    }
    int read(T, Args...)(ref T x, auto ref Args args) {
        if (!readSingle(x)) return 0;
        static if (args.length == 0) {
            return 1;
        } else {
            return 1 + read(args);
        }
    }
}

unittest {
    import std.path : buildPath;
    import std.file : tempDir;
    import std.algorithm : equal;
    import std.stdio : File;
    string fileName = buildPath(tempDir, "kyuridenanmaida.txt");
    auto fout = File(fileName, "w");
    fout.writeln("1 2 3");
    fout.writeln("ab cde");
    fout.writeln("1.0 1.0 2.0");
    fout.close;
    Scanner sc = new Scanner(File(fileName, "r"));
    int a;
    int[2] b;
    char[2] c;
    string d;
    double e;
    double[] f;
    sc.read(a, b, c, d, e, f);
    assert(a == 1);
    assert(equal(b[], [2, 3]));
    assert(equal(c[], "ab"));
    assert(equal(d, "cde"));
    assert(e == 1.0);
    assert(equal(f, [1.0, 2.0]));
}


Measured memory_bytes: 1916